This script accompanies the paper *Inequalities à la Pólya for the Aharonov—Bohm eigenvalues of the disk* by N. Filonov, M. Levitin, I. Polterovich, and D. A. Sher, see Proposition 3.8

For rigorous verified rational approximations proofs of some inequalities appearing in the text see the end of the script

Preliminaries

In[3]:=
```
h[x_] := 1 / Pi (Sqrt[1 - x^2] - x ArcCos[x]);
G[λ_, z_] := λ h[z / λ];
Wdisk[d_, λ_] := (λ / 2)^d Gamma[1 + d / 2]^(-2);
```

Verified approximations

In[6]:=
```
CosUp[x_] := 1 - x^2/2 + x^4/24 - x^6/720 + x^8/40320 - x^10/3628800 + x^12/479001600;
CosDown[x_] := 1 - x^2/2 + x^4/24 - x^6/720 + x^8/40320 - x^10/3628800 + x^12/479001600 - x^14/87178291200;
VerifiedQUpArcCos[x_, ϵ_] := Module[{ac0, ac},
   ac0 = ArcCos[x];
   If[MatchQ[ac0, _Rational] || IntegerQ[ac0], ac = ac0,
    ac = Rationalize[ac0 + 2 ϵ, ϵ];
    If[CosUp[ac] ≥ x, Message[VerifiedQUpArcCos::Error, x], ac]];
   (* ac=Rationalize[ArcCos[x]+2ϵ,ϵ]; *)
   ac
  ];
VerifiedQUpArcCos::Error = "ArcCosUp of argument `1` is wrong!";
```

```
VerifiedQDownArcCos[x_, ϵ_] := Module[{ac0, ac},
    ac0 = ArcCos[x];
    If[MatchQ[ac0, _Rational] || IntegerQ[ac0], ac = ac0,
     ac = Rationalize[ac0 - 2 ϵ, ϵ];
     If[CosDown[ac] ≤ x, Message[VerifiedQDownArcCos::Error, x], ac]];
    ac
  ];
VerifiedQDownArcCos::Error = "ArcCosDown of argument `1` is wrong!";
VerifiedQUpSqrt[x_, ϵ_] := Module[{ac0, ac},
    ac0 = Sqrt[x];
    If[MatchQ[ac0, _Rational] || IntegerQ[ac0], ac = ac0,
     ac = Rationalize[ac0 + 2 ϵ, ϵ];
     If[ac^2 < x, Message[VerifiedQUpSqrt::Error, x], ac]];
    ac
  ];
VerifiedQUpSqrt::Error = "SqrtUp of argument `1` is wrong!";
VerifiedQDownSqrt[x_, ϵ_] := Module[{ac0, ac},
    ac0 = Sqrt[x];
    If[MatchQ[ac0, _Rational] || IntegerQ[ac0], ac = ac0,
     ac = Rationalize[ac0 - 2 ϵ, ϵ];
     If[ac^2 > x, Message[VerifiedQDownSqrt::Error, x], ac]];
    ac
  ];
VerifiedQDownSqrt::Error = "SqrtDown of argument `1` is wrong!";
VerifiedQUpRoot[x_, d_, ϵ_] := Module[{ac0, ac},
    ac0 = x^(1 / d);
    If[MatchQ[ac0, _Rational] || IntegerQ[ac0], ac = ac0,
     ac = Rationalize[ac0 + 2 ϵ, ϵ];
     If[ac^d < x, Message[VerifiedQUpRoot::Error, x], ac]];
    ac
  ];
VerifiedQUpRoot::Error = "RootUp `2` of argument `1` is wrong!";
VerifiedQDownRoot[x_, d_, ϵ_] := Module[{ac0, ac},
    ac0 = x^(1 / d);
    If[MatchQ[ac0, _Rational] || IntegerQ[ac0], ac = ac0,
     ac = Rationalize[ac0 - 2 ϵ, ϵ];
     If[ac^d > x, Message[VerifiedQDownRoot::Error, x], ac]];
    ac
  ];
VerifiedQDownSqrt::Error = "RootDown `2` of argument `1` is wrong!";
```

Algorithm

In[20]:=
```
PiUp[ϵ_] := 3 VerifiedQUpArcCos[1 / 2, ϵ];
PiDown[ϵ_] := 3 VerifiedQDownArcCos[1 / 2, ϵ];
GQDown[λ_, z_, ϵ_] := If[z > λ, 0,
   (VerifiedQDownSqrt[λ^2 - z^2, ϵ] - z VerifiedQUpArcCos[z / λ, ϵ]) / PiUp[ϵ]];
WdiskQUp[d_, λ_, ϵ_ ] := Wdisk[d, λ] /. Pi → PiDown[ϵ];
adisk[d_, ϵ_] := WdiskQUp[d, λ, ϵ] / λ^d;
PABNeuQDown[λ_, ϵ_] := Total[Table[Floor[GQDown[λ, m + 1 / 2, ϵ] + 3 / 4] +
     Floor[GQDown[λ, m + 1, ϵ] + 3 / 4], {m, 0, Ceiling[λ] - 1}]];
```

In[26]:=
```
StepCheckAB[λ_, ϵ_] := Module[{e, λnew},
   e = PABNeuQDown[λ, ϵ] - WdiskQUp[2, λ, ϵ];
   If[e ≤ 0, Print["Process stopped with e=", e]; Return[]];
   λnew = VerifiedQDownRoot[λ^2 + e / adisk[2, ϵ], 2, ϵ];
   {λ, λ // N, e, e // N, λnew, λnew // N}
  ];
RunChecksAB[Λstart_, Λend_, ϵ_, printstep_] := Module[{Λ, nI, data, step},
   data = {};
   nI = 1;
   Λ = Λstart;
   Do[
    step = StepCheckAB[Λ, ϵ];
    If[printstep, Print[step]];
    AppendTo[data, step];
    If[step[[5]] ≤ Λ, Break[]];
    Λ = step[[5]];
    If[Λ ≥ Λend , Break[]];
    ,
    {nI, 1, 100 000}
   ];
   step = StepCheckAB[Λ, ϵ];
   If[printstep, Print[step]];
   AppendTo[data, step];
   data
  ];
```

Actual Calculation

In[28]:=
```
Λ20short = 5 / 2;
Λ21short = 9;
data2dshort = RunChecksAB[Λ20short, Λ21short, 10^(-3), True];
tb2short = Table[{j, data2dshort[[j, 1]],
    data2dshort[[j, 3]], data2dshort[[j, 5]] - data2dshort[[j, 1]]},
   {j, 1, Length[data2dshort]}] // TableForm
```

$$\left\{\frac{5}{2}, 2.5, \frac{7}{16}, 0.4375, \frac{65}{23}, 2.82609\right\}$$

$$\left\{\frac{65}{23}, 2.82609, \frac{2123}{2116}, 1.00331, \frac{45}{13}, 3.46154\right\}$$

$$\left\{\frac{45}{13}, 3.46154, \frac{679}{676}, 1.00444, 4, 4.\right\}$$

$$\left\{4, 4., 1, 1., \frac{76}{17}, 4.47059\right\}$$

$$\left\{\frac{76}{17}, 4.47059, \frac{290}{289}, 1.00346, \frac{142}{29}, 4.89655\right\}$$

$$\left\{\frac{142}{29}, 4.89655, \frac{846}{841}, 1.00595, \frac{164}{31}, 5.29032\right\}$$

$$\left\{\frac{164}{31}, 5.29032, \frac{964}{961}, 1.00312, \frac{164}{29}, 5.65517\right\}$$

$$\left\{\frac{164}{29}, 5.65517, \frac{1686}{841}, 2.00476, \frac{196}{31}, 6.32258\right\}$$

$$\left\{\frac{196}{31}, 6.32258, \frac{1928}{961}, 2.00624, \frac{187}{27}, 6.92593\right\}$$

$$\left\{\frac{187}{27}, 6.92593, \frac{5855}{2916}, 2.00789, \frac{202}{27}, 7.48148\right\}$$

$$\left\{\frac{202}{27}, 7.48148, \frac{1463}{729}, 2.00686, 8, 8.\right\}$$

$$\left\{8, 8., 2, 2., \frac{246}{29}, 8.48276\right\}$$

$$\left\{\frac{246}{29}, 8.48276, \frac{1691}{841}, 2.0107, \frac{313}{35}, 8.94286\right\}$$

$$\left\{\frac{313}{35}, 8.94286, \frac{14\,731}{4900}, 3.00633, \frac{374}{39}, 9.58974\right\}$$

$$\left\{\frac{374}{39}, 9.58974, \frac{4577}{1521}, 3.0092, \frac{418}{41}, 10.1951\right\}$$

Out[31]//TableForm=

1	$\frac{5}{2}$	$\frac{7}{16}$	$\frac{15}{46}$
2	$\frac{65}{23}$	$\frac{2123}{2116}$	$\frac{190}{299}$
3	$\frac{45}{13}$	$\frac{679}{676}$	$\frac{7}{13}$
4	4	1	$\frac{8}{17}$
5	$\frac{76}{17}$	$\frac{290}{289}$	$\frac{210}{493}$
6	$\frac{142}{29}$	$\frac{846}{841}$	$\frac{354}{899}$
7	$\frac{164}{31}$	$\frac{964}{961}$	$\frac{328}{899}$
8	$\frac{164}{29}$	$\frac{1686}{841}$	$\frac{600}{899}$
9	$\frac{196}{31}$	$\frac{1928}{961}$	$\frac{505}{837}$
10	$\frac{187}{27}$	$\frac{5855}{2916}$	$\frac{5}{9}$
11	$\frac{202}{27}$	$\frac{1463}{729}$	$\frac{14}{27}$
12	8	2	$\frac{14}{29}$
13	$\frac{246}{29}$	$\frac{1691}{841}$	$\frac{467}{1015}$
14	$\frac{313}{35}$	$\frac{14731}{4900}$	$\frac{883}{1365}$
15	$\frac{374}{39}$	$\frac{4577}{1521}$	$\frac{968}{1599}$

Verifying some inequalities

$\sqrt{3}+\frac{6}{5\pi}-\frac{5\pi}{9}>0$, page 10

In[33]:=
```
VerifiedQDownSqrt[3, 10^(-3)] + 6 / (5 PiUp[10^(-3)]) - 5 PiUp[10^(-3)] / 9
```

Out[33]=

$$\frac{395}{1092}$$

$\frac{5\pi}{6}<2\sqrt{2}$, page 10

In[34]:=
```
5 / 6 PiUp[10^(-3)] - 2 VerifiedQDownSqrt[2, 10^(-3)]
```

Out[34]=

$$-\frac{27}{136}$$

$q_2\left(\frac{783}{1000}\right) > q_1\left(\frac{783}{1000}\right) > 3$, page 14

In[35]:=
```
c0 = 783 / 1000;
q2Down = 1 / (3481 / 2500 - 4 / PiUp[10^(-5)]);
q1Up = PiUp[10^(-5)] / (4 (VerifiedQDownSqrt[1 - c0^2, 10^(-5)] -
        c0 VerifiedQUpArcCos[c0, 10^(-5)]));
q1Down = PiDown[10^(-5)] / (4 (VerifiedQUpSqrt[1 - c0^2, 10^(-5)] -
        c0 VerifiedQDownArcCos[c0, 10^(-5)]));
q2Down - q1Up
q1Down - 3
```

Out[39]=
$$\frac{21\,748\,540\,846\,500}{91\,065\,923\,628\,253}$$

Out[40]=
$$\frac{593\,657}{115\,381}$$